This *Mathematica* notebook accompanies the article "Zeilberger's Holonomic Ansatz for Pfaffians" by Masao Ishikawa and Christoph Koutschan (Proceedings of the ISSAC 2012, pp. 227-233).
The packages that are required to reproduce the computations can be downloaded freely from the following webpage: http://www.risc.jku.at/research/combinat/software/

```
<< Guess.m
<< HolonomicFunctions.m
<< Hyper.m
```

```
Guess  Package by Manuel Kauers — © RISC Linz — V 0.41 2011-11-22
```

HolonomicFunctions package by Christoph Koutschan, RISC-Linz, Version 1.6 (12.04.2012)
⟶ Type ?HolonomicFunctions for help

Motzkin Number Pfaffian (Theorem 2)

Evaluate the Pfaffian $\mathrm{Pf}\big(a(i,j)_{1\le i,j\le 2n}\big)$ where $a(i,j) = (j-i)\,M(i+j-3)$ with $M(n)$ denoting the Motzkin numbers: $M(n) = \sum_{k=0}^{n} \frac{1}{k+1}\binom{n}{2k}\binom{2k}{k}$. The evaluation of this Pfaffian was conjectured in the paper "Pfaffian decomposition and a Pfaffian analogue of q-Catalan Hankel determinants" (Masao Ishikawa, Hiroyuki Tagawa, Jiang Zeng, arXiv:1009.2004), see Formula (6.3) there.

```
MotzkinNumber[0 | 1] := 1; MotzkinNumber[n_Integer /; n > 1] :=
 MotzkinNumber[n] = (3 * (n - 1) * MotzkinNumber[n - 2] + (2 n + 1) * MotzkinNumber[n - 1]) / (n + 2);
```

```
Clear[a, aij];
a[i_Integer, j_Integer] := (j - i) * MotzkinNumber[i + j - 3];
aij = DFiniteTimes[Annihilator[j - i, {S[i], S[j]}],
  First[CreativeTelescoping[Binomial[i + j - 3, 2 k] * CatalanNumber[k], S[k] - 1, {S[i], S[j]}]]]
```

$$\Big\{(-1+i-j)\,S_i + (-1-i+j)\,S_j,\ \left(-i+i^3+j-2ij-i^2j+2j^2-ij^2+j^3\right)S_j^2 + \left(-2i+5i^2-2i^3+2j-2ij+2i^2j-3j^2+2ij^2-2j^3\right)S_j + \left(12-24i+15i^2-3i^3+12j-12ij+3i^2j-3j^2+3ij^2-3j^3\right)\Big\}$$

Now we guess an implicit description (linear recurrences) for the auxiliary function $c_{2n,i}$. We set the option AdditionalEquations to Infinity, which means that all data is used for guessing. Thus we can be sure that the resulting recurrences hold for all values in the given array. In other words, it doesn't make a difference whether we produce values for $c_{2n,i}$ from its definition (recurrences plus initial values) or from the data array.

```
Timing[
 dim = 30; data = {};
 Do[
  matrix = Table[a[i, j], {j, 2 n - 1}, {i, 2 n - 1}];
  vec = First[LinSolveQ[matrix]];
  vec = PadRight[vec / Last[vec], 2 * dim];
  AppendTo[data, vec];
  , {n, dim}];

 (* The parameters of the following commands are the result of trial and error. *)
 guess1 = GuessMultRE[data, {f[n, i], f[n, i + 1], f[n + 1, i], f[n, i + 2], f[n + 1, i + 1]},
    {n, i}, 3, StartPoint → {1, 1}, AdditionalEquations → Infinity];
 guess2 = GuessMultRE[data, {f[n, i], f[n, i + 1], f[n + 1, i], f[n, i + 2], f[n + 2, i]},
    {n, i}, 6, StartPoint → {1, 1}, AdditionalEquations → Infinity];
 guess3 = GuessMultRE[data, {f[n, i], f[n, i + 1], f[n + 1, i], f[n, i + 2], f[n, i + 3]},
    {n, i}, 4, StartPoint → {1, 1}, AdditionalEquations → Infinity];
 c2ni = OreGroebnerBasis[ToOrePolynomial[First /@ {guess1, guess2, guess3}, f[n, i]]];
]
```

{15.605, Null}

```
Factor[c2ni]
```

$$\{i\,(-1+2n)\,(-3+4n)\,S_n S_i - 24\,i\,(1+i)\,n\,S_i^2 - (-1+i-2n)\,(-1+2n)\,(-3+4n)\,S_n - i\,(1-2n+16\,i\,n+8n^2)\,S_i + (-1+i+2n)\,(1-2n+8\,i\,n-8n^2),$$
$$n\,(-1+2n)\,(2-i+2n)\,(3-i+2n)\,(-3+4n)\,(1+4n)\,S_n^2 - 36\,i\,(1+i)\,n\,(1+n)\,(1+2n)\,(1+4n)\,S_i^2 - (-1+2n)\,(-3+4n)\,(-i+i^2+10n-16\,i\,n+8\,i^2 n+6n^2-16\,i\,n^2+8\,i^2 n^2-20n^3-16n^4)\,S_n - 6\,i\,(1+4\,i)\,n\,(1+n)\,(1+2n)\,(1+4n)\,S_i + (1+n)\,(-1+i+2n)\,(1+4n)\,(i+8n+6\,i\,n-24n^2+32\,i\,n^2-32n^3),$$
$$18\,i\,(1+i)\,(2+i)\,n\,S_i^3 + 3\,i\,(1+i)\,(3+14\,i-12n)\,n\,S_i^2 + (-1+i-2n)\,(i-2n)\,(-1+2n)\,(-3+4n)\,S_n + 2\,i\,n\,(-2-4\,i+7\,i^2-3n-12\,i\,n)\,S_i - (-1+i+2n)\,(i-2n-3\,i\,n+10\,i^2 n+4n^2-24\,i\,n^2+16n^3)\}$$

```
UnderTheStaircase[c2ni]
```

$$\{1,\ S_i,\ S_n,\ S_i^2\}$$

```
Factor[FGLM[c2ni, Lexicographic]]
```

$$\{18\,(1+i)\,(2+i)\,(3+i)\,S_i^4 + 3\,(1+i)\,(2+i)\,(11+8\,i)\,S_i^3 - (1+i)\,(7+7\,i+4\,i^2+24n-24n^2)\,S_i^2 + (-i-9\,i^2-8\,i^3-4n-16\,i\,n+4n^2+16\,i\,n^2)\,S_i + (-1+2\,i)\,(1+i-2n)\,(-1+i+2n),$$
$$(-1+i-2n)\,(i-2n)\,(-1+2n)\,(-3+4n)\,S_n + 18\,i\,(1+i)\,(2+i)\,n\,S_i^3 + 3\,i\,(1+i)\,(3+14\,i-12n)\,n\,S_i^2 + 2\,i\,n\,(-2-4\,i+7\,i^2-3n-12\,i\,n)\,S_i - (-1+i+2n)\,(i-2n-3\,i\,n+10\,i^2 n+4n^2-24\,i\,n^2+16n^3)\}$$

The recurrences that generate the annihilating ideal c2ni do not have any singularities, and thus we have to specify only 4 initial values, corresponding to the monomials under the stairs:

```
AnnihilatorSingularities[c2ni, {1, 1}]
```

{{{i → 1, n → 1}, True}, {{i → 1, n → 2}, True}, {{i → 2, n → 1}, True}, {{i → 3, n → 1}, True}}

■ Boundary conditions for $c_{2n,i}$

We show that $c_{2n,i} = 0$ for $i \leq 0$ and for $i \geq 2n$. This will be useful for the summations later ("natural boundaries").

```
Factor[c2ni[[2]]]
```

```
-n (-1 + 2 n) (2 - i + 2 n) (3 - i + 2 n) (-3 + 4 n) (1 + 4 n) Sn^2 + 36 i (1 + i) n (1 + n) (1 + 2 n) (1 + 4 n) Si^2 +
  (-1 + 2 n) (-3 + 4 n) (-i + i^2 + 10 n - 16 i n + 8 i^2 n + 6 n^2 - 16 i n^2 + 8 i^2 n^2 - 20 n^3 - 16 n^4) Sn +
  6 i (1 + 4 i) n (1 + n) (1 + 2 n) (1 + 4 n) Si -
  (1 + n) (-1 + i + 2 n) (1 + 4 n) (i + 8 n + 6 i n - 24 n^2 + 32 i n^2 - 32 n^3)
```

Provided with the appropriate initial conditions ($c_{2,0} = c_{4,0} = 0$), we see that this recurrence produces zeros on the line $i = 0$, since the terms S_i^2 and S_i vanish. Similarly for $i = -1$, since the term S_i^2 still vanishes (provided that $c_{2,-1} = c_{4,-1} = 0$). Because of these two zero rows, it is clear that everything beyond them (i.e., for $i < -1$) must be zero as well. The following computation shows that setting the initial conditions to 0 is compatible with the recurrences:

```
test = ApplyOreOperator[c2ni, c[n, i]]; Union[
 Flatten[Table[test, {n, 1, 5}, {i, -5, 5}] /. c[_, i_?NonPositive] → 0 /. c[a__] ⧴ data[[a]]]]
```

```
{0}
```

Now recall the leading coefficient and the support of the first defining recurrence of $c_{2n,i}$:

```
{Factor[LeadingCoefficient[c2ni[[1]]]], Support[c2ni[[1]]]}
```

```
{i (-1 + 2 n) (-3 + 4 n), {Sn Si, Si^2, Sn, Si, 1}}
```

Since this coefficient does not vanish for any integer point in the area $n \geq 2$ and $i \geq 2n$, we can use this recurrence to produce the values of $c_{2n,i}$ in this area. The support of this recurrence indicates that we need only to show that $c_{2n,2n} = 0$.

```
Factor[LeadingCoefficient[First[DFiniteSubstitute[c2ni, {i → 2 n}]]]]
```

```
648 (1 + n) (2 + n) (3 + 2 n) (5 + 2 n) (-3 + 4 n) (1 + 4 n) (5 + 4 n) (10 + 191 n + 814 n^2 + 560 n^3)
```

The leading coefficient having no relevant singularities and the initial values being zero (by construction) show this. With the same argument it is clear that also $c_{2,i} = 0$.

```
Factor[LeadingCoefficient[First[DFiniteSubstitute[c2ni, {n → 1}]]]]
```

```
18 (2 + i) (3 + i)
```

■ Identity (1)

We compute an annihilating operator for $c_{2n,2n-1}$. Its leading coefficient has no nonnegative integer roots, and it has the operator $(S_n - 1)$ as a right factor. Therefore it annihilates any constant sequence.

```
Factor[diag = DFiniteSubstitute[c2ni, {i → 2 n - 1}]]
```

$$\{648\,(2+n)\,(3+n)\,(3+2\,n)\,(5+2\,n)\,(9+4\,n)\,(-163\,269-1\,249\,360\,n-2\,766\,651\,n^2+318\,598\,n^3+9\,388\,056\,n^4+12\,424\,768\,n^5+6\,117\,888\,n^6+1\,003\,520\,n^7)\,S_n^4-9\,(2+n)\,(3+2\,n)\,(-4\,055\,570\,397-36\,787\,892\,604\,n-112\,871\,472\,099\,n^2-91\,332\,046\,174\,n^3+250\,055\,729\,072\,n^4+709\,690\,617\,792\,n^5+790\,485\,373\,440\,n^6+477\,893\,039\,616\,n^7+163\,799\,322\,624\,n^8+29\,872\,783\,360\,n^9+2\,247\,884\,800\,n^{10})\,S_n^3+2\,(7+4\,n)\,(-10\,807\,720\,002-108\,882\,519\,192\,n-389\,769\,960\,693\,n^2-456\,703\,818\,406\,n^3+748\,802\,397\,776\,n^4+3\,180\,051\,863\,416\,n^5+4\,691\,486\,928\,384\,n^6+3\,859\,379\,040\,096\,n^7+1\,918\,451\,296\,896\,n^8+571\,364\,924\,416\,n^9+93\,646\,249\,984\,n^{10}+6\,470\,696\,960\,n^{11})\,S_n^2-(3+4\,n)\,(7+4\,n)\,(-807\,182\,010-9\,027\,892\,737\,n-31\,775\,261\,328\,n^2-11\,738\,389\,051\,n^3+170\,887\,753\,992\,n^4+429\,645\,397\,612\,n^5+483\,258\,827\,168\,n^6+299\,664\,483\,136\,n^7+105\,291\,703\,808\,n^8+19\,565\,342\,720\,n^9+1\,485\,209\,600\,n^{10})\,S_n+2\,(-1+4\,n)^2\,(1+4\,n)\,(3+4\,n)\,(7+4\,n)\,(25\,073\,550+137\,581\,164\,n+291\,607\,399\,n^2+319\,599\,462\,n^3+198\,403\,416\,n^4+70\,206\,016\,n^5+13\,142\,528\,n^6+1\,003\,520\,n^7)\}$$

```
OreReduce[diag, Annihilator[1, S[n]]]
```

$\{0\}$

The four initial values are 1 by construction and therefore $c_{2n,2n-1} = 1$ for all n.

■ Identity (2)

Compute a creative telescoping operator for the left-hand side.

```
alg = OreAlgebra[S[n], S[i], S[j]];
Timing[ByteCount[smnd = DFiniteTimes[
    ToOrePolynomial[Append[aij, S[n] - 1], alg], ToOrePolynomial[Append[c2ni, S[j] - 1], alg]]]]
```

{0.896056, 482 488}

```
Timing[fct = FindCreativeTelescoping[smnd, S[i] - 1];]
```

{24.9856, Null}

```
Factor[First[fct]]
```

$$\{-j\,(j+2\,n)\,(-3+4\,n)\,S_n+n\,(-1-j+2\,n)\,(1+4\,n)\,S_j+j\,(j-n)\,(1+4\,n),\; -(2+j-2\,n)\,(j+2\,n)\,S_j^2+(1+j)\,(1+2\,j)\,S_j+3\,j\,(1+j)\}$$

These two operators annihilate the left-hand-side expression of identity (2). If we consider them as the defining recurrences for a bivariate sequence (in n and j) then we need the following initial values to fill the area $1 \le j < 2\,n$:

```
AnnihilatorSingularities[fct[[1]], {1, 1}, Assumptions → j < 2 n]
```

{{{j → 1, n → 1}, True}, {{j → 1, n → 2}, True}, {{j → 2, n → 2}, True}}

```
ReleaseHold[Hold[Sum[data[[n, i]] * a[i, j], {i, 1, 2 n - 1}]] /. (First /@ %)]
```

{0, 0, 0}

This concludes the proof of (2).

Remark: note that the above reasoning is about the maximal possible area. If one tries to extend it further, the first step being $j = 2\,n$, the second recurrence found above breaks down:

```
Collect[ApplyOreOperator[fct[[1, 2]], f[n, j]] /. j → j - 2, f[__], Factor]
```

```
3 (-2 + j) (-1 + j) f[n, -2 + j] + (-1 + j) (-3 + 2 j) f[n, -1 + j] - (j - 2 n) (-2 + j + 2 n) f[n, j]
```

```
AnnihilatorSingularities[fct[[1]], {1, 1}, Assumptions → j ≤ 2 n]
```

```
{{{j → 1, n → 1}, True}, {{j → 2, n → 1}, True}, {{j → 2, n → 2}, True},
 {{j → 2 + 2 C[1], n → 1 + C[1]}, C[1] ∈ Integers && C[1] ≥ 0}}
```

Indeed, the values for $j = 2n$ are nonzero as we demonstrate in the next section.

▪ Identity (3)

```
ai2n = DFiniteSubstitute[aij, {j → 2 n}, Algebra → OreAlgebra[S[n], S[i]]]
```

$$\{(i + 2 i^2 + i^3 - 2 n - 4 i n - 2 i^2 n - 4 i n^2 + 8 n^3) S_n + (-2 i + 5 i^2 - 2 i^3 + 4 n - 4 i n + 4 i^2 n - 12 n^2 + 8 i n^2 - 16 n^3) S_i + (-12 + 9 i^2 - 3 i^3 + 24 n - 24 i n + 6 i^2 n + 12 n^2 + 12 i n^2 - 24 n^3),$$
$$(i + 2 i^2 + i^3 - 2 n - 4 i n - 2 i^2 n - 4 i n^2 + 8 n^3) S_i^2 + (2 i - 3 i^2 - 2 i^3 - 4 n - 4 i n + 4 i^2 n + 20 n^2 + 8 i n^2 - 16 n^3) S_i + (12 + 12 i - 3 i^2 - 3 i^3 - 48 n - 24 i n + 6 i^2 n + 60 n^2 + 12 i n^2 - 24 n^3)\}$$

Compute recurrences for the summand of identity (3):

```
Timing[ByteCount[smnd = DFiniteTimes[c2ni, ai2n]]]
```

```
{8.70454, 2 593 640}
```

Compute a creative telescoping operator in order to deal with the summation in identity (3):

```
Timing[ByteCount /@ (fct = FindCreativeTelescoping[smnd, S[i] - 1])]
```

```
{436.103, {2328, 512 464}}
```

```
fct = << "fct_6.3.m";
```

Check the correctness of the creative telescoping operator (show that it is a member of the left ideal generated by the recurrences of smnd):

```
{{principalPart}, {{deltaPart}}} = fct;
Timing[OreReduce[principalPart + (S[i] - 1) ** deltaPart, smnd]]
```

```
{8.28852, 0}
```

Hence the ratio $r(n) = \frac{b(2n)}{b(2n-2)}$ satisfies the following recurrence (actually from $n = 1$ on, since the Pfaffian evaluation is also true for $n = 0$):

```
rec = ApplyOreOperator[Factor[principalPart], r[n]]
```

$$9 n (1 + 2 n) (1 + 4 n) (8 + 7 n) r[n] - (-3 + 4 n) (30 + 346 n + 687 n^2 + 350 n^3) r[1 + n] + 2 (-3 + 4 n) (1 + 4 n) (3 + 4 n) (1 + 7 n) r[2 + n]$$

Together with the initial values we get a closed form for this quotient.

```
Table[Sum[data[[n, i]] * a[i, 2 n], {i, 1, 2 n - 1}], {n, 1, 2}]
```

```
{1, 5}
```

```
RSolve[{rec == 0, r[1] == 1, r[2] == 5}, r[n], n]
```

```
{{r[n] → -3 + 4 n}}
```

It follows that $b(2\,n) = \prod_{k=1}^{n} \frac{b(2\,k)}{b(2\,k-2)} = \prod_{k=1}^{n} (4\,k-3) = \prod_{k=0}^{n-1} (4\,k+1)$.

Central Delannoy Number Pfaffian (Theorem 3)

Evaluate the Pfaffian $\mathrm{Pf}\left(a\,(i,\,j)_{1\le i,j\le 2\,n}\right)$ where $a(i,\,j) = (j-i)\,D(i+j-3)$ with $D(n)$ denoting the central Delannoy numbers: $D(n) = \sum_{k=0}^{n} \binom{n}{k}\binom{n+k}{k}$. The evaluation of this Pfaffian was conjectured in the paper "Pfaffian decomposition and a Pfaffian analogue of q-Catalan Hankel determinants" (Masao Ishikawa, Hiroyuki Tagawa, Jiang Zeng, arXiv:1009.2004), see Formula (6.4) there.

```
CentralDelannoyNumber[0] := 1;
CentralDelannoyNumber[1] := 3;
CentralDelannoyNumber[n_Integer /; n > 1] := CentralDelannoyNumber[n] =
   (-(n - 1) * CentralDelannoyNumber[n - 2] + 3 * (2 n - 1) * CentralDelannoyNumber[n - 1]) / n;
```

```
Clear[a, aij];
a[i_Integer, j_Integer] := (j - i) * CentralDelannoyNumber[i + j - 3];
aij = Annihilator[
   (j - i) * Sum[Binomial[i + j - 3, k] * Binomial[i + j - 3 + k, k], {k, 0, i + j - 3}], {S[i], S[j]}];
Factor[
 aij]
```

$$\left\{(-1+i-j)\,S_i + (-1-i+j)\,S_j,\ (-1+i-j)\,(i-j)\,(-1+i+j)\,S_j^2 - 3\,(-2+i-j)\,(i-j)\,(-3+2\,i+2\,j)\,S_j + (-2+i-j)\,(-1+i-j)\,(-2+i+j)\right\}$$

Now we guess an implicit description (linear recurrences) for the auxiliary function $c_{2\,n,i}$. We set the option AdditionalEquations to Infinity, which means that all data is used for guessing. Thus we can be sure that the resulting recurrences hold for all values in the given array. In other words, it doesn't make a difference whether we produce values for $c_{2\,n,i}$ from its definition (recurrences plus initial values) or from the data array.

```
Timing[
 dim = 30; data = {};
 Do[
  matrix = Table[a[i, j], {j, 2 n - 1}, {i, 2 n - 1}];
  vec = First[LinSolveQ[matrix]];
  vec = PadRight[vec / Last[vec], 2 * dim];
  AppendTo[data, vec];
  , {n, dim}];

 (* The parameters of the following commands are the result of trial and error. *)
 guess1 = GuessMultRE[data, {f[n, i], f[n, i + 1], f[n + 1, i], f[n, i + 2], f[n + 1, i + 1]},
    {n, i}, 4, StartPoint → {1, 1}, AdditionalEquations → Infinity];
 guess2 = GuessMultRE[data, {f[n, i], f[n, i + 1], f[n + 1, i], f[n, i + 2], f[n + 2, i]},
    {n, i}, 8, StartPoint → {1, 1}, AdditionalEquations → Infinity];
 guess3 = GuessMultRE[data, {f[n, i], f[n, i + 1], f[n + 1, i], f[n, i + 2], f[n, i + 3]},
    {n, i}, 5, StartPoint → {1, 1}, AdditionalEquations → Infinity];
 c2ni = OreGroebnerBasis[ToOrePolynomial[First /@ {guess1, guess2, guess3}, f[n, i]]];
]
```

```
{16.653, Null}
```

```
Factor[c2ni]
```

$$\{3\, i\, (-1+n)\, (-1+2 n)\, (-5+4 n)\, S_n S_i + 8\, i\, (1+i)\, n\, (-1+2 n)\, S_i^2 - (-1+i-2 n)\, (-1+n)\, (-1+2 n)\, (-5+4 n)\, S_n - 6\, i\, n\, \left(11-8 i-22 n+16 i n+8 n^2\right) S_i - 2 n\, (-3+i+2 n)\, \left(-3+4 i-2 n-8 i n+8 n^2\right),$$
$$(-3+i-2 n)\, (-2+i-2 n)\, (-1+n)\, (-1+2 n)^2\, (1+2 n)\, (-5+4 n)\, (-1+4 n)\, S_n^2 - 8\, i\, (1+i)\, n\, (1+n)\, (-1+2 n)\, (1+2 n)^2\, (-1+4 n)\, S_i^2 - 2\, (-1+n)\, (1+n)\, (-1+2 n)\, (-5+4 n)\, \left(21-8 i-2 n-60 n^2-64 i n^2+32 i^2 n^2+8 n^3+160 n^4\right) S_n + 12\, i\, (-1+4 i)\, n\, (1+n)\, (-1+2 n)\, (1+2 n)^2\, (-1+4 n)\, S_i + 4\, (1+n)\, (1+2 n)^2\, (-3+i+2 n)\, (-1+4 n)\, \left(-18+9 i+57 n-16 i n-62 n^2+4 i n^2+24 n^3\right),$$
$$-6\, i\, (1+i)\, (2+i)\, n\, (-1+2 n)\, S_i^3 + i\, (1+i)\, n\, (-1+2 n)\, (-27+70 i+4 n)\, S_i^2 - 2\, (-1+i-2 n)\, (i-2 n)\, (-1+n)\, (-1+2 n)\, (-5+4 n)\, S_n - 6\, i\, n\, (-1+2 n)\, \left(13-62 i+35 i^2-n+4 i n\right) S_i + n\, (-3+i+2 n)\, \left(39 i-34 i^2-24 n-14 i n+68 i^2 n-16 n^2-96 i n^2+64 n^3\right)\}$$

The recurrences become slightly nicer when we transform the Gröbner basis to the lexicographic monomial order (however, we will work with the original c2ni, which corresponds to degree lexicographic order):

```
Factor[FGLM[c2ni, Lexicographic]]
```

$$\{2\, (1+i)\, (2+i)\, (3+i)\, S_i^4 - 3\, (1+i)\, (2+i)\, (5+8 i)\, S_i^3 + (1+i)\, \left(-31+19 i+76 i^2+16 n-8 n^2\right) S_i^2 - 3\, \left(4-13 i-9 i^2+8 i^3-8 n+32 i n+4 n^2-16 i n^2\right) S_i + (-3+2 i)\, (1+i-2 n)\, (-3+i+2 n),$$
$$2\, (-1+i-2 n)\, (i-2 n)\, (-1+n)\, (-1+2 n)\, (-5+4 n)\, S_n + 6\, i\, (1+i)\, (2+i)\, n\, (-1+2 n)\, S_i^3 - i\, (1+i)\, n\, (-1+2 n)\, (-27+70 i+4 n)\, S_i^2 + 6\, i\, n\, (-1+2 n)\, \left(13-62 i+35 i^2-n+4 i n\right) S_i - n\, (-3+i+2 n)\, \left(39 i-34 i^2-24 n-14 i n+68 i^2 n-16 n^2-96 i n^2+64 n^3\right)\}$$

```
UnderTheStaircase[c2ni]
```

$$\{1,\, S_i,\, S_n,\, S_i^2\}$$

We determine the points for which the recurrences in c2ni cannot be applied. We need to include the initial conditions for those points explicitly.

```
AnnihilatorSingularities[c2ni, {1, 1}]
```

```
{{{i → 1, n → 1}, True}, {{i → 1, n → 2}, True}, {{i → 1, n → 3}, True},
 {{i → 2, n → 1}, True}, {{i → 2, n → 2}, True}, {{i → 3, n → 1}, True}, {{i → 3, n → 2}, True}}
```

■ Boundary conditions for $c_{2n,i}$

We show that $c_{2n,i} = 0$ for $i \le 0$ and for $i \ge 2n$. This will be useful for the summations later ("natural boundaries").

```
Factor[c2ni[[2]]]
```

$$(-3+i-2 n)\, (-2+i-2 n)\, (-1+n)\, (-1+2 n)^2\, (1+2 n)\, (-5+4 n)\, (-1+4 n)\, S_n^2 - 8\, i\, (1+i)\, n\, (1+n)\, (-1+2 n)\, (1+2 n)^2\, (-1+4 n)\, S_i^2 - 2\, (-1+n)\, (1+n)\, (-1+2 n)\, (-5+4 n)\, \left(21-8 i-2 n-60 n^2-64 i n^2+32 i^2 n^2+8 n^3+160 n^4\right) S_n + 12\, i\, (-1+4 i)\, n\, (1+n)\, (-1+2 n)\, (1+2 n)^2\, (-1+4 n)\, S_i + 4\, (1+n)\, (1+2 n)^2\, (-3+i+2 n)\, (-1+4 n)\, \left(-18+9 i+57 n-16 i n-62 n^2+4 i n^2+24 n^3\right)$$

Provided with the appropriate initial conditions ($c_{2,0} = c_{4,0} = 0$), we see that this recurrence produces zeros on the line $i = 0$, since the terms $S_i{}^2$ and S_i vanish. Similarly for $i = -1$, since the term $S_i{}^2$ still vanishes (provided that $c_{2,-1} = c_{4,-1} = 0$). Because of these two zero rows, it is clear that everything beyond them (i.e., for $i < -1$) must be zero as well. The following computation shows that setting the initial conditions to 0 is compatible with the recurrences:

```
test = ApplyOreOperator[c2ni, c[n, i]]; Union[
 Flatten[Table[test, {n, 1, 5}, {i, -5, 5}] /. c[_, i_?NonPositive] → 0 /. c[a__] :→ data[[a]]]]
```

$$\{0\}$$

Now recall the leading coefficient and the support of the first defining recurrence of $c_{2n,i}$:

```
{Factor[LeadingCoefficient[c2ni[[1]]]], Support[c2ni[[1]]]}
```

$$\left\{3\, i\, (-1+n)\, (-1+2n)\, (-5+4n),\ \left\{S_n S_i,\ S_i^2,\ S_n,\ S_i,\ 1\right\}\right\}$$

Since this coefficient does not vanish for any integer point in the area $n \geq 2$ and $i \geq 2n$, we can use this recurrence to produce the values of $c_{2n,i}$ in this area. The support of this recurrence indicates that we need only to show that $c_{2n,2n} = 0$.

```
Factor[LeadingCoefficient[First[DFiniteSubstitute[c2ni, {i → 2 n}]]]]
```

$$4\, (-1+n)\, n\, (1+n)\, (1+2n)\, (3+2n)\, (5+2n)\, (-5+4n)\, (-1+4n)\, (3+4n)\, \left(4-59n+190n^2+336n^3\right)$$

The leading coefficient having no relevant singularities and the initial values being zero (by construction) show this. With the same argument it is clear that also $c_{2,i} = 0$.

```
Factor[LeadingCoefficient[First[DFiniteSubstitute[c2ni, {n → 1}]]]]
```

$$6\, i\, (1+i)\, (2+i)$$

Identity (1)

We compute an annihilating operator for $c_{2n,2n-1}$. Its leading coefficient has no nonnegative integer roots, and it has the operator $(S_n - 1)$ as a right factor. Therefore it annihilates any constant sequence.

```
Factor[diag = DFiniteSubstitute[c2ni, {i → 2 n - 1}]]
```

$$\begin{aligned}\{&4\, (1+n)\, (2+n)\, (3+2n)\, (5+2n)^2\, (7+4n) \\ &\quad (50\,900 - 863\,022\, n + 4\,845\,391\, n^2 + 414\,996\, n^3 - 82\,083\,920\, n^4 + 50\,982\,272\, n^5 + 395\,635\,376\, n^6 - \\ &\qquad 99\,949\,760\, n^7 - 712\,214\,400\, n^8 - 91\,284\,480\, n^9 + 389\,265\,408\, n^{10} + 151\,732\,224\, n^{11})\, S_n^4 - \\ &\ (1+n)\, (3+2n)^2\, (-2\,451\,046\,500 - 7\,297\,862\,370\, n + 235\,749\,825\,109\, n^2 - 52\,228\,728\,380\, n^3 - \\ &\qquad 3\,854\,392\,971\,320\, n^4 - 1\,371\,957\,867\,328\, n^5 + 19\,486\,466\,742\,288\, n^6 + 18\,945\,985\,923\,648\, n^7 - \\ &\qquad 28\,244\,533\,036\,800\, n^8 - 48\,432\,778\,639\,360\, n^9 - 7\,800\,420\,883\,456\, n^{10} + 25\,473\,012\,805\,632\, n^{11} + \\ &\qquad 19\,978\,256\,793\,600\, n^{12} + 5\,935\,562\,293\,248\, n^{13} + 645\,772\,345\,344\, n^{14})\, S_n^3 + \\ &\ 3\, (1+2n)\, (5+4n)\, (-65\,210\,903\,100 + 160\,231\,012\,560\, n + 2\,925\,401\,846\,821\, n^2 - 2\,360\,520\,995\,393\, n^3 - \\ &\qquad 40\,915\,166\,897\,074\, n^4 - 19\,439\,672\,623\,544\, n^5 + 191\,488\,486\,600\,560\, n^6 + 261\,048\,493\,548\,336\, n^7 - \\ &\qquad 191\,037\,865\,860\,000\, n^8 - 604\,188\,020\,398\,848\, n^9 - 323\,531\,076\,363\,520\, n^{10} + 201\,099\,237\,828\,608\, n^{11} + \\ &\qquad 328\,823\,156\,527\,104\, n^{12} + 167\,070\,778\,785\,792\, n^{13} + 39\,345\,870\,864\,384\, n^{14} + 3\,631\,862\,513\,664\, n^{15})\, S_n^2 - \\ &\ 2\, (-1+2n)\, (1+4n)\, (5+4n)\, (94\,178\,342\,040 - 379\,115\,774\,291\, n - 4\,071\,942\,782\,050\, n^2 + \\ &\qquad 283\,702\,644\,303\, n^3 + 47\,038\,371\,379\,798\, n^4 + 93\,154\,230\,866\,368\, n^5 - 18\,534\,676\,587\,616\, n^6 - \\ &\qquad 251\,614\,448\,451\,280\, n^7 - 283\,378\,414\,266\,784\, n^8 - 32\,412\,725\,874\,304\, n^9 + 182\,417\,276\,284\,160\, n^{10} + \\ &\qquad 174\,991\,159\,944\,192\, n^{11} + 75\,184\,616\,779\,776\, n^{12} + 16\,128\,217\,792\,512\, n^{13} + 1\,399\,578\,034\,176\, n^{14})\, S_n + \\ &\ 16\, (-3+2n)\, (-3+4n)^2\, (-1+4n)\, (1+4n)\, (5+4n)\, (6\,530\,985 + 653\,245\,388\, n + 6\,492\,966\,459\, n^2 + \\ &\quad 28\,792\,131\,156\, n^3 + 73\,069\,633\,600\, n^4 + 117\,135\,168\,992\, n^5 + 123\,932\,110\,704\, n^6 + \\ &\quad 87\,699\,576\,640\, n^7 + 41\,018\,985\,600\, n^8 + 12\,146\,641\,920\, n^9 + 2\,058\,319\,872\, n^{10} + 151\,732\,224\, n^{11})\}\end{aligned}$$

```
OreReduce[diag, Annihilator[1, S[n]]]
```

$$\{0\}$$

The four initial values are 1 by construction and therefore $c_{2n,2n-1} = 1$ for all n.

Identity (2)

Compute a creative telescoping operator for the left-hand side.

```
alg = OreAlgebra[S[n], S[i], S[j]];
Timing[ByteCount[smnd = DFiniteTimes[
    ToOrePolynomial[Append[aij, S[n] - 1], alg], ToOrePolynomial[Append[c2ni, S[j] - 1], alg]]]]
```

```
{1.00806, 642120}
```

```
Timing[ByteCount /@ (fct = FindCreativeTelescoping[smnd, S[i] - 1])]
```

```
{25.6496, {5016, 92744}}
```

```
Factor[First[fct]]
```

$$\{-2\,j\,(-1+2\,n)\,(-2+j+2\,n)\,(-5+4\,n)\,S_n-3\,(1+j-2\,n)\,(-3+2\,n)\,(-1+2\,n)\,(-1+4\,n)\,S_j+ j\,(-3+2\,n)\,(-1+4\,n)\,(-9+4\,j+10\,n),\ (2+j-2\,n)\,(-2+j+2\,n)\,S_j^2-3\,(1+j)\,(-1+2\,j)\,S_j+j\,(1+j)\}$$

These two operators annihilate the left-hand-side expression of identity (2). If we consider them as the defining recurrences for a bivariate sequence (in n and j) then we need the following initial values to fill the area $1 \le j < 2\,n$:

```
AnnihilatorSingularities[fct[[1]], {1, 1}, Assumptions → j < 2 n]
```

```
{{{j → 1, n → 1}, True}, {{j → 1, n → 2}, True}, {{j → 2, n → 2}, True}}
```

```
ReleaseHold[Hold[Sum[data[[n, i]] * a[i, j], {i, 1, 2 n - 1}]] /. (First /@ %)]
```

```
{0, 0, 0}
```

This concludes the proof of (2).

Remark: note that the above reasoning is about the maximal possible area. If one tries to extend it further, the first step being $j = 2\,n$, the second recurrence found above breaks down:

```
Collect[ApplyOreOperator[fct[[1, 2]], f[n, j]] /. j → j - 2, f[__], Factor]
```

$$(-2+j)\,(-1+j)\,f[n,-2+j]-3\,(-1+j)\,(-5+2\,j)\,f[n,-1+j]+(j-2\,n)\,(-4+j+2\,n)\,f[n,j]$$

```
AnnihilatorSingularities[fct[[1]], {1, 1}, Assumptions → j ≤ 2 n]
```

```
{{{j → 1, n → 1}, True}, {{j → 2, n → 1}, True}, {{j → 2, n → 2}, True},
 {{j → 2 + 2 C[1], n → 1 + C[1]}, C[1] ∈ Integers && C[1] ≥ 0}}
```

Indeed, the values for $j = 2\,n$ are nonzero as we demonstrate in the next section.

■ Identity (3)

```
ai2n = DFiniteSubstitute[aij, {j → 2 n}, Algebra → OreAlgebra[S[n], S[i]]]
```

$$\{(-i+i^3+2\,n+4\,i\,n-2\,i^2\,n-8\,n^2-4\,i\,n^2+8\,n^3)\,S_n+ (-18\,i+21\,i^2-6\,i^3+36\,n-36\,i\,n+12\,i^2\,n-12\,n^2+24\,i\,n^2-48\,n^3)\,S_i+ (4-3\,i^2+i^3-8\,n+8\,i\,n-2\,i^2\,n-4\,n^2-4\,i\,n^2+8\,n^3),$$
$$(-i+i^3+2\,n+4\,i\,n-2\,i^2\,n-8\,n^2-4\,i\,n^2+8\,n^3)\,S_i^2+ (18\,i-3\,i^2-6\,i^3-36\,n-36\,i\,n+12\,i^2\,n+84\,n^2+24\,i\,n^2-48\,n^3)\,S_i+ (-4-4\,i+i^2+i^3+16\,n+8\,i\,n-2\,i^2\,n-20\,n^2-4\,i\,n^2+8\,n^3)\}$$

Compute recurrences for the summand of identity (3):

```
Timing[ByteCount[smnd = DFiniteTimes[c2ni, ai2n]]]
```

```
{10.7927, 3 525 240}
```

Compute a creative telescoping operator in order to deal with the summation in identity (3):

```
Timing[ByteCount /@ (fct = FindCreativeTelescoping[smnd, S[i] - 1])]
```

```
{658.849, {3024, 745 376}}
```

```
fct = << "fct_6.4.m";
```

Check the correctness of the creative telescoping operator (show that it is a member of the left ideal generated by the recurrences of smnd):

```
{{principalPart}, {{deltaPart}}} = fct;
Timing[OreReduce[principalPart + (S[i] - 1) ** deltaPart, smnd]]
```

```
{11.4767, 0}
```

Hence the ratio $r(n) = \frac{b(2n)}{b(2n-2)}$ satisfies the following recurrence (for $n \geq 2$, since the evaluation is conjectured to be true for $n \geq 1$):

```
rec = ApplyOreOperator[Factor[principalPart], r[n]]
```

$$4 n (-3+2 n) (1+2 n) (3+2 n) (-1+4 n) (11+14 n) r[n] - (-1+2 n) (3+2 n) (-5+4 n) \left(-72-59 n+1416 n^2+1820 n^3\right) r[1+n] + 2 (1+2 n)^2 (-5+4 n) (-1+4 n) (1+4 n) (-3+14 n) r[2+n]$$

Indeed, the recurrence is valid for all n>1, but it does not hold for n=1:

```
Table[rec, {n, 1, 10}] /. r[n_] :> Sum[data[[n, i]] * a[i, 2 n], {i, 1, 2 n - 1}]
```

```
{4500, 0, 0, 0, 0, 0, 0, 0, 0, 0}
```

Together with the initial values we get a closed form for this quotient. We use Marko Petkovsek's implementation of his algorithm Hyper to find the hypergeometric solutions of this recurrence. There is only one, but it turns out that it is exactly the one that we are looking for (two initial values match!).

```
{hyp} = Hyper[rec, r[n]]
```

$$\left\{\frac{4 (-3+2 n) (1+2 n) (-1+4 n)}{(-1+2 n)^2 (-5+4 n)}\right\}$$

```
Table[Sum[data[[n, i]] * a[i, 2 n], {i, 1, 2 n - 1}], {n, 1, 10}]
```

$$\left\{1, 72, \frac{1120}{3}, \frac{9856}{5}, \frac{69120}{7}, \frac{428032}{9}, \frac{2449408}{11}, \frac{13271040}{13}, \frac{69074944}{15}, \frac{348651520}{17}\right\}$$

```
RSolve[{r[n + 1] / r[n] == hyp, r[2] == 72, r[3] == 1120 / 3}, r[n], n]
```

$$\left\{\left\{r[n] \to \frac{2^{-1+2 n} (-1+2 n) (-5+4 n)}{-3+2 n}\right\}\right\}$$

```
Table[%[[1, 1, 2]], {n, 1, 10}]
```

$$\left\{2, 72, \frac{1120}{3}, \frac{9856}{5}, \frac{69120}{7}, \frac{428032}{9}, \frac{2449408}{11}, \frac{13271040}{13}, \frac{69074944}{15}, \frac{348651520}{17}\right\}$$

The solution matches except for the first value. This means that $r_1 = 1$ and $r_n = \frac{2^{2n-1}(2n-1)(4n-5)}{2n-3}$ for $n \geq 2$.

It follows that $b(2n) = \prod_{k=1}^{n} \frac{b(2k)}{b(2k-2)} = \prod_{k=1}^{n} r_k = \prod_{k=1}^{n} \frac{2^{2k-1}(2k-1)(4k-5)}{2k-3} = 2^{n^2} \prod_{k=1}^{n} \frac{(2k-1)(4k-5)}{2k-3} = -2^{n^2}(2n-1)\prod_{k=1}^{n}(4k-5) = 2^{n^2}(2n-1)\prod_{k=1}^{n-1}(4k-1)$.

Narayana Polynomial Pfaffian (Theorem 4)

Evaluate the Pfaffian $\text{Pf}\left(a(i, j)_{1\leq i,j\leq 2n}\right)$ where $a(i, j) = (j-i)\, N(i+j-2, x)$ with $N(n, x)$ denoting the n-th Narayana polynomial: $N(n, x) = \sum_{k=0}^{n} \frac{1}{n} \binom{n}{k} \binom{n}{k-1} x^k$. The evaluation of this Pfaffian was conjectured in the paper "Pfaffian decomposition and a Pfaffian analogue of q-Catalan Hankel determinants" (Masao Ishikawa, Hiroyuki Tagawa, Jiang Zeng, arXiv:1009.2004), see Formula (6.6) there.

```
NarayanaN[1, x_] := x;
NarayanaN[2, x_] := x * (x + 1);
NarayanaN[n_Integer /; n > 2, x_] := NarayanaN[n, x] =
   Expand[
    (- (x - 1) ^2 * (n - 2) * NarayanaN[n - 2, x] + (x + 1) * (2 n - 1) * NarayanaN[n - 1, x]) / (n + 1)];

Clear[a, aij];
a[i_Integer, j_Integer] := (j - i) * NarayanaN[i + j - 2, x];
aij =
  Annihilator[(j - i) * Sum[1 / (i + j - 2) * Binomial[i + j - 2, k] * Binomial[i + j - 2, k - 1] * x^k,
      {k, 0, i + j - 2}], {S[i], S[j]}];
Factor[
 aij]
```

```
{(-1 + i - j) Si + (-1 - i + j) Sj, (-1 + i - j) (i - j) (1 + i + j) Sj^2 -
  (-2 + i - j) (i - j) (-1 + 2 i + 2 j) (1 + x) Sj + (-2 + i - j) (-1 + i - j) (-2 + i + j) (-1 + x)^2}
```

Now we guess an implicit description (linear recurrences) for the auxiliary function $c_{2n,i}$.

```
Timing[
 dim = 26; data = {};
 Do[
  matrix = Table[a[i, j], {j, 2 n - 1}, {i, 2 n - 1}];
  vec = First[LinSolveUniv[matrix, x]];
  vec = PadRight[vec / Last[vec], 2 * dim];
  AppendTo[data, vec];
  , {n, dim}];

 (* The parameters of the following commands are the result of trial and error. *)
 guess1 = GuessMultRE[data, {f[n, i], f[n, i + 1], f[n + 1, i], f[n, i + 2], f[n + 1, i + 1]}, {n, i},
    3, StartPoint → {1, 1}, Constraints → 2 i ≤ n, AdditionalEquations → Infinity];
 guess2 = GuessMultRE[data, {f[n, i], f[n, i + 1], f[n + 1, i], f[n, i + 2], f[n + 2, i]},
    {n, i}, 6, StartPoint → {1, 1}, Constraints → 2 i ≤ n, AdditionalEquations → Infinity];
 guess3 = GuessMultRE[data, {f[n, i], f[n, i + 1], f[n + 1, i], f[n, i + 2], f[n, i + 3]},
    {n, i}, 4, StartPoint → {1, 1}, Constraints → 2 i ≤ n, AdditionalEquations → Infinity];
 c2ni = OreGroebnerBasis[ToOrePolynomial[First /@ {guess1, guess2, guess3}, f[n, i]]];
]
```

```
{1511.44, Null}
```

```
c2ni = << "c2ni_6.6.m";
```

```
Factor[c2ni]
```

```
{i (-1 + 2 n) (-3 + 4 n) (1 + x) Sn Si + 8 i (1 + i) n (-1 + x)^2 x Si^2 - (-1 + i - 2 n) (-1 + 2 n) (-3 + 4 n) Sn -
   i (1 - 2 n + 16 i n + 8 n^2) x (1 + x) Si + (-1 + i + 2 n) (1 - 2 n + 8 i n - 8 n^2) x,
  n (-1 + 2 n) (2 - i + 2 n) (3 - i + 2 n) (-3 + 4 n) (1 + 4 n) Sn^2 -
   4 i (1 + i) n (1 + n) (1 + 2 n) (1 + 4 n) (-1 + x)^4 x Si^2 -
   (-1 + 2 n) (-3 + 4 n) (2 n + 14 n^2 + 28 n^3 + 16 n^4 - i x + i^2 x + 12 n x - 16 i n x + 8 i^2 n x +
       20 n^2 x - 16 i n^2 x + 8 i^2 n^2 x + 8 n^3 x + 2 n x^2 + 14 n^2 x^2 + 28 n^3 x^2 + 16 n^4 x^2) Sn +
   2 i (1 + 4 i) n (1 + n) (1 + 2 n) (1 + 4 n) (-1 + x)^2 x (1 + x) Si + (1 + n) (-1 + i + 2 n) (1 + 4 n)
    x (-2 n - 4 i n + 4 n^2 - 8 i n^2 + 16 n^3 + i x + 6 n x + 2 i n x - 20 n^2 x + 24 i n^2 x - 16 n^3 x -
      2 n x^2 - 4 i n x^2 + 4 n^2 x^2 - 8 i n^2 x^2 + 16 n^3 x^2), -2 i (1 + i) (2 + i) n (-1 + x)^4 (1 + x) Si^3 +
   i (1 + i) n (-1 + x)^2 (3 + 6 i + 4 n + 6 x + 20 i x - 8 n x + 3 x^2 + 6 i x^2 + 4 n x^2) Si^2 -
   (-1 + i - 2 n) (i - 2 n) (-1 + 2 n) (-3 + 4 n) Sn -
   2 i n (1 + x) (3 i^2 + n + 4 i n - 2 x - 4 i x + 10 i^2 x - 2 n x - 8 i n x + 3 i^2 x^2 + n x^2 + 4 i n x^2) Si +
   (-1 + i + 2 n) (-i n + 2 i^2 n + i x - 2 n x - 4 i n x + 12 i^2 n x + 4 n^2 x - 24 i n^2 x + 16 n^3 x - i n x^2 + 2 i^2 n x^2)}
```

We determine the (finitely many) points for which the recurrences in c2ni cannot be applied. There are for such points in general, and some more in the special case $x = -1$. So for the moment, let's assume that $x \neq -1$.

```
AnnihilatorSingularities[c2ni, {1, 1}]
```

```
{{{i → 1, n → 1}, True}, {{i → 1, n → 2}, True}, {{i → 2, n → 1}, True},
 {{i → 3, n → 1}, True}, {{i → 2, n → 1, x → -1}, True}, {{i → 2, n → 2, x → -1}, True},
 {{i → 3, n → 1, x → -1}, True}, {{i → 3, n → 2, x → -1}, True}}
```

■ Boundary conditions for $c_{2n,i}$

We show that $c_{2n,i} = 0$ for $i \leq 0$ and for $i \geq 2n$. This will be useful for the summations later ("natural boundaries").

```
Factor[c2ni[[2]]]
```

```
n (-1 + 2 n) (2 - i + 2 n) (3 - i + 2 n) (-3 + 4 n) (1 + 4 n) Sn^2 -
 4 i (1 + i) n (1 + n) (1 + 2 n) (1 + 4 n) (-1 + x)^4 x Si^2 -
 (-1 + 2 n) (-3 + 4 n) (2 n + 14 n^2 + 28 n^3 + 16 n^4 - i x + i^2 x + 12 n x - 16 i n x + 8 i^2 n x +
     20 n^2 x - 16 i n^2 x + 8 i^2 n^2 x + 8 n^3 x + 2 n x^2 + 14 n^2 x^2 + 28 n^3 x^2 + 16 n^4 x^2) Sn +
 2 i (1 + 4 i) n (1 + n) (1 + 2 n) (1 + 4 n) (-1 + x)^2 x (1 + x) Si +
 (1 + n) (-1 + i + 2 n) (1 + 4 n) x (-2 n - 4 i n + 4 n^2 - 8 i n^2 + 16 n^3 + i x + 6 n x +
    2 i n x - 20 n^2 x + 24 i n^2 x - 16 n^3 x - 2 n x^2 - 4 i n x^2 + 4 n^2 x^2 - 8 i n^2 x^2 + 16 n^3 x^2)
```

Provided with the appropriate initial conditions ($c_{2,0} = c_{4,0} = 0$), we see that this recurrence produces zeros on the line $i = 0$, since the terms S_i^2 and S_i vanish. Similarly for $i = -1$, since the term S_i^2 still vanishes (provided that $c_{2,-1} = c_{4,-1} = 0$). Because of these two zero rows, it is clear that everything beyond them (i.e., for $i < -1$) must be zero as well. The following computation shows that setting the initial conditions to 0 is compatible with the recurrences:

```
test = ApplyOreOperator[c2ni, c[n, i]]; Union[Flatten[
  Expand[Table[test, {n, 1, 5}, {i, -5, 5}] /. c[_, i_?NonPositive] → 0 /. c[a__] ⧴ data[[a]]]]]
```

```
{0}
```

Now recall the leading coefficient and the support of the first defining recurrence of $c_{2n,i}$:

```
{Factor[LeadingCoefficient[c2ni[[1]]]], Support[c2ni[[1]]]}
```

```
{i (-1 + 2 n) (-3 + 4 n) (1 + x), {Sn Si, Si^2, Sn, Si, 1}}
```

Since this coefficient does not vanish for any integer point in the area $n \geq 2$ and $i \geq 2n$ (again assuming $x \neq -1$), we can use this recurrence to produce the values of $c_{2n,i}$ in this area. The support of this recurrence indicates that we need only to show that $c_{2n,2n} = 0$.

```
Factor[LeadingCoefficient[First[DFiniteSubstitute[c2ni, {i → 2 n}]]]]
```

```
8 (1 + n) (2 + n) (3 + 2 n) (5 + 2 n) (-3 + 4 n) (1 + 4 n) (5 + 4 n) (-1 + x)^8
 (9 n + 26 n^2 + 16 n^3 - 10 x - 182 n x - 788 n^2 x - 544 n^3 x + 9 n x^2 + 26 n^2 x^2 + 16 n^3 x^2)
```

```
CylindricalDecomposition[Implies[n > 1 && x < -1, % > 0], {n, x}]
```

```
True
```

The leading coefficient having no relevant singularities (assuming $x < -1$) and the initial values being zero (by construction) show this. With the same argument it is clear that also $c_{2,i} = 0$.

```
Factor[LeadingCoefficient[First[DFiniteSubstitute[c2ni, {n → 1}]]]]
```

```
2 (2 + i) (3 + i) (-1 + x)^4
```

Identity (1)

We compute an annihilating operator for $c_{2n,2n-1}$. Its leading coefficient has no nonnegative integer roots (for $x < -1$), and it has the operator $(S_n - 1)$ as a right factor. Therefore it annihilates any constant sequence.

```
Timing[ByteCount[diag = First[DFiniteSubstitute[c2ni, {i → 2 n - 1}]]]]
```

```
{37.6344, 182 280}
```

```
OreReduce[diag, Annihilator[1, S[n]]]
```

```
0
```

```
Factor[LeadingCoefficient[diag]]
```

```
8 (2 + n) (3 + n) (3 + 2 n) (5 + 2 n) (9 + 4 n) (-1 + x)^8
 (-3600 n - 18 840 n^2 - 30 088 n^3 + 544 n^4 + 50 720 n^5 + 56 320 n^6 + 25 088 n^7 + 4096 n^8 + 240 x + 44 491 n x +
   344 881 n^2 x + 623 558 n^3 x - 400 944 n^4 x - 2 248 096 n^5 x - 2 459 136 n^6 x - 1 123 328 n^7 x - 188 416 n^8 x -
   21 117 x^2 - 338 790 n x^2 - 2 239 327 n^2 x^2 - 3 561 830 n^3 x^2 + 5 462 048 n^4 x^2 + 21 087 424 n^5 x^2 +
   22 707 200 n^6 x^2 + 10 533 376 n^7 x^2 + 1 806 336 n^8 x^2 + 147 513 x^3 + 1 457 250 n x^3 + 5 559 065 n^2 x^3 +
   5 183 466 n^3 x^3 - 18 809 424 n^4 x^3 - 53 876 832 n^5 x^3 - 55 231 488 n^6 x^3 - 25 525 248 n^7 x^3 - 4 431 872 n^8 x^3 -
   200 622 x^4 - 1 472 842 n x^4 - 2 922 166 n^2 x^4 + 981 628 n^3 x^4 + 6 231 168 n^4 x^4 - 3 392 960 n^5 x^4 -
   15 313 920 n^6 x^4 - 10 713 088 n^7 x^4 - 2 244 608 n^8 x^4 + 147 513 x^5 + 1 457 250 n x^5 + 5 559 065 n^2 x^5 +
   5 183 466 n^3 x^5 - 18 809 424 n^4 x^5 - 53 876 832 n^5 x^5 - 55 231 488 n^6 x^5 - 25 525 248 n^7 x^5 - 4 431 872 n^8 x^5 -
   21 117 x^6 - 338 790 n x^6 - 2 239 327 n^2 x^6 - 3 561 830 n^3 x^6 + 5 462 048 n^4 x^6 + 21 087 424 n^5 x^6 +
   22 707 200 n^6 x^6 + 10 533 376 n^7 x^6 + 1 806 336 n^8 x^6 + 240 x^7 + 44 491 n x^7 + 344 881 n^2 x^7 + 623 558 n^3 x^7 -
   400 944 n^4 x^7 - 2 248 096 n^5 x^7 - 2 459 136 n^6 x^7 - 1 123 328 n^7 x^7 - 188 416 n^8 x^7 - 3600 n x^8 -
   18 840 n^2 x^8 - 30 088 n^3 x^8 + 544 n^4 x^8 + 50 720 n^5 x^8 + 56 320 n^6 x^8 + 25 088 n^7 x^8 + 4096 n^8 x^8)
```

```
CylindricalDecomposition[Implies[n > 1 && x < -1, % > 0], {n, x}]
```

```
True
```

The four initial values are 1 by construction and therefore $c_{2n,2n-1} = 1$ for all n.

```
Table[data[[n, 2 n - 1]], {n, 4}]
```

```
{1, 1, 1, 1}
```

Identity (2)

Compute a creative telescoping operator for the left-hand side.

```
alg = OreAlgebra[S[n], S[i], S[j]];
Timing[ByteCount[smnd = DFiniteTimes[
    ToOrePolynomial[Append[aij, S[n] - 1], alg], ToOrePolynomial[Append[c2ni, S[j] - 1], alg]]]]
```

```
{1.96412, 1733712}
```

```
Timing[ByteCount /@ (fct = FindCreativeTelescoping[smnd, S[i] - 1])]
```

```
{81.7651, {5472, 284640}}
```

```
Factor[First[fct]]
```

$$\{-j\,(j+2n)\,(-3+4n)\,S_n + n\,(-1-j+2n)\,(1+4n)\,(1+x)\,S_j + j\,(1+4n)\,(n+j\,x+n\,x^2),\ (2+j-2n)\,(j+2n)\,S_j^2 - (1+j)\,(1+2j)\,(1+x)\,S_j + j\,(1+j)\,(-1+x)^2\}$$

These two operators annihilate the left-hand-side expression of identity (2). If we consider them as the defining recurrences for a bivariate sequence (in n and j) then we need the following initial values to fill the area $1 \le j < 2n$:

```
AnnihilatorSingularities[fct[[1]], {1, 1}, Assumptions → j < 2 n]
```

```
{{{j → 1, n → 1}, True}, {{j → 1, n → 2}, True}, {{j → 2, n → 2}, True}}
```

```
Expand[ReleaseHold[Hold[Sum[data[[n, i]] * a[i, j], {i, 1, 2 n - 1}]] /. (First /@ %)]]
```

```
{0, 0, 0}
```

This concludes the proof of (2).

Remark: note that the above reasoning is about the maximal possible area. If one tries to extend it further, the first step being $j = 2n$, the second recurrence found above breaks down:

```
Collect[ApplyOreOperator[fct[[1, 2]], f[n, j]] /. j → j - 2, f[__], Factor]
```

$$(-2+j)\,(-1+j)\,(-1+x)^2\,f[n,-2+j] - (-1+j)\,(-3+2j)\,(1+x)\,f[n,-1+j] + (j-2n)\,(-2+j+2n)\,f[n,j]$$

```
AnnihilatorSingularities[fct[[1]], {1, 1}, Assumptions → j ≤ 2 n]
```

```
{{{j → 1, n → 1}, True}, {{j → 2, n → 1}, True}, {{j → 2, n → 2}, True},
  {{j → 2 + 2 C[1], n → 1 + C[1]}, C[1] ∈ Integers && C[1] ≥ 0}}
```

Indeed, the values for $j = 2n$ are nonzero as we demonstrate in the next section.

Identity (3)

```
ai2n = DFiniteSubstitute[aij, {j → 2 n}, Algebra → OreAlgebra[S[n], S[i]]]
```

$$
\begin{aligned}
&\{(i + 2 i^2 + i^3 - 2 n - 4 i n - 2 i^2 n - 4 i n^2 + 8 n^3)\, S_n + \\
&\quad (-2 i + 5 i^2 - 2 i^3 + 4 n - 4 i n + 4 i^2 n - 12 n^2 + 8 i n^2 - 16 n^3 - 2 i x + \\
&\qquad 5 i^2 x - 2 i^3 x + 4 n x - 4 i n x + 4 i^2 n x - 12 n^2 x + 8 i n^2 x - 16 n^3 x)\, S_i + \\
&\quad (4 - 3 i^2 + i^3 - 8 n + 8 i n - 2 i^2 n - 4 n^2 - 4 i n^2 + 8 n^3 - 8 x + 6 i^2 x - 2 i^3 x + 16 n x - 16 i n x + 4 i^2 n x + \\
&\qquad 8 n^2 x + 8 i n^2 x - 16 n^3 x + 4 x^2 - 3 i^2 x^2 + i^3 x^2 - 8 n x^2 + 8 i n x^2 - 2 i^2 n x^2 - 4 n^2 x^2 - 4 i n^2 x^2 + 8 n^3 x^2), \\
&\quad (i + 2 i^2 + i^3 - 2 n - 4 i n - 2 i^2 n - 4 i n^2 + 8 n^3)\, S_i^2 + (2 i - 3 i^2 - 2 i^3 - 4 n - 4 i n + 4 i^2 n + 20 n^2 + \\
&\qquad 8 i n^2 - 16 n^3 + 2 i x - 3 i^2 x - 2 i^3 x - 4 n x - 4 i n x + 4 i^2 n x + 20 n^2 x + 8 i n^2 x - 16 n^3 x)\, S_i + \\
&\quad (-4 - 4 i + i^2 + i^3 + 16 n + 8 i n - 2 i^2 n - 20 n^2 - 4 i n^2 + 8 n^3 + 8 x + 8 i x - 2 i^2 x - \\
&\qquad 2 i^3 x - 32 n x - 16 i n x + 4 i^2 n x + 40 n^2 x + 8 i n^2 x - 16 n^3 x - 4 x^2 - 4 i x^2 + \\
&\qquad i^2 x^2 + i^3 x^2 + 16 n x^2 + 8 i n x^2 - 2 i^2 n x^2 - 20 n^2 x^2 - 4 i n^2 x^2 + 8 n^3 x^2)\}
\end{aligned}
$$

Compute recurrences for the summand of identity (3):

```
Timing[ByteCount[smnd = DFiniteTimes[c2ni, ai2n]]]
```

{110.883, 36 712 216}

Compute a creative telescoping operator in order to deal with the summation in identity (3).
With evaluation/interpolation it took us 1955s to compute such an operator.

```
fct = << "fct_6.6.m";
```

Check the correctness of the creative telescoping operator (show that it is a member of the left ideal generated by the recurrences of smnd):

```
{{principalPart}, {{deltaPart}}} = fct;
Timing[OreReduce[principalPart + (S[i] - 1) ** deltaPart, smnd]]
```

{203.253, 0}

Hence the ratio $r(n) = \frac{b(2n)}{b(2n-2)}$ satisfies the following recurrence (for $n \geq 1$, since the evaluation of the Pfaffian holds for $n \geq 0$):

```
rec = ApplyOreOperator[Factor[principalPart], r[n]]
```

$$
\begin{aligned}
&n\,(1 + 2 n)\,(1 + 4 n)\,(-1 + x)^4 x^2\,(1 + n - 7 x - 6 n x + x^2 + n x^2)\, r[n] - \\
&\quad (-3 + 4 n)\,(n + 3 n^2 + 2 n^3 - 11 n x - 32 n^2 x - 20 n^3 x + 65 n x^2 + 165 n^2 x^2 + 94 n^3 x^2 - 30 x^3 - 294 n x^3 - \\
&\qquad 560 n^2 x^3 - 280 n^3 x^3 + 65 n x^4 + 165 n^2 x^4 + 94 n^3 x^4 - 11 n x^5 - 32 n^2 x^5 - 20 n^3 x^5 + n x^6 + 3 n^2 x^6 + 2 n^3 x^6) \\
&\quad r[1 + n] + 2\,(-3 + 4 n)\,(1 + 4 n)\,(3 + 4 n)\,(n - x - 6 n x + n x^2)\, r[2 + n]
\end{aligned}
$$

Together with the initial values we get a closed form for this quotient. We use Marko Petkovsek's implementation of his algorithm Hyper to find the hypergeometric solutions of this recurrence. There is only one, but it turns out that it is exactly the one that we are looking for (two initial values match!).

```
{hyp} = Hyper[rec, r[n]]
```

$$\left\{\frac{(1 + 4 n)\, x^2}{-3 + 4 n}\right\}$$

```
Table[Together[Sum[data[[n, i]] * a[i, 2 n], {i, 1, 2 n - 1}]], {n, 1, 10}]
```

$$\{x,\ 5 x^3,\ 9 x^5,\ 13 x^7,\ 17 x^9,\ 21 x^{11},\ 25 x^{13},\ 29 x^{15},\ 33 x^{17},\ 37 x^{19}\}$$

```
RSolve[{r[n + 1] / r[n] == hyp, r[1] == x, r[2] == 5 x^3}, r[n], n]
```

$$\left\{\left\{r[n] \to \frac{(-3 + 4\, n)\, \left(x^2\right)^n}{x}\right\}\right\}$$

Mathematica can also solve this recurrence directly, but this takes much longer:

```
Timing[RSolve[{rec == 0, r[1] == x, r[2] == 5 x^3}, r[n], n]]
```

$$\left\{18.7612, \left\{\left\{r[n] \to \frac{(-3 + 4\, n)\, \left(x^2\right)^n}{x}\right\}\right\}\right\}$$

```
Table[%[[2, 1, 1, 2]], {n, 1, 10}]
```

$\{x, 5\, x^3, 9\, x^5, 13\, x^7, 17\, x^9, 21\, x^{11}, 25\, x^{13}, 29\, x^{15}, 33\, x^{17}, 37\, x^{19}\}$

It follows that $b(2\, n) = \prod_{k=1}^{n} \frac{b(2\, k)}{b(2\, k-2)} = \prod_{k=1}^{n} r_k = \prod_{k=1}^{n} \left((4\, k-3)\, x^{2\, k-1}\right) = x^{n^2} \prod_{k=1}^{n} (4\, k-3) = x^{n^2} \prod_{k=0}^{n-1} (4\, k+1)$.

Final remark: in some steps we have assumed that x is a real number smaller than -1. Hence for now the evaluation is proven only for $x < -1$. But for specific n, the Pfaffian is a polynomial in x (of a certain degree), as well as the right-hand side evaluation. Thus their difference is a polynomial in x which is zero for all $x < -1$. By the fundamental theorem of algebra it follows that this polynomial is identically zero, and therefore the evaluation of the Pfaffian is true for all complex numbers x.